# COMP1022Q Review Questions Week 3

*These practice questions, as well as the answers, are released for the content of each week. These are just for you to practice. You don't have to hand them in. There's no marks for doing them, but it will help your understanding if you go through them.*

Q1)

Here is a worksheet.

| | A | B | C |
|---|---|---|---|
| 1 | **Member's name:** | Tina | |
| 2 | **Going to participate?** | Yes | |
| 3 | | | |
| 4 | **Member** | **Confirmed to participate?** | |
| 5 | Ann | Yes | |
| 6 | Ben | No | |
| 7 | John | Yes | |
| 8 | Tina | Yes | |
| 9 | Tom | No | |
| 10 | Sally | Yes | |
| 11 | Mary | No | |

Cell A5 to cell B11 contains a list of members' confirmation of whether they will participate in a particular event or not. A formula has been entered in cell B2 to check whether a particular member is going to participate or not. The worksheet is used by entering the member's name in cell B1. Then cell B2 shows 'Yes' if that person is going to participate or 'No' if that person is not going to participate.

You need to complete the formula in cell B2 shown below.

```
=VLOOKUP(B1, _____ , _____ , _____ )
```

Q2)

Here is a worksheet.

| | A | B | C |
|---|---|---|---|
| 1 | **Family Members** | **Money In wallet** | |
| 2 | Ann | $ 200.00 | |
| 3 | Ben | $ 100.00 | |
| 4 | Dad | $ 20.00 | |
| 5 | Mom | $ 500.00 | |

Cell B2 to cell B5 have been displayed using the currency format. You can ignore the dollar signs shown at the start of those cells when you answer this question, because the dollar signs are not actually inside the content of the cells, they are just displayed by Excel.

Here is a formula.

```
=IF(VLOOKUP("Dad", A2:B5, 2, FALSE) < MAX(B2:B5), "Poor Dad!", "Rich Dad!")
```

What is the result of the above formula? You don't have to explain what the formula does. Just write down what the formula shows (the result).

**Answer:** _____

Q3)

Here is a worksheet.

| | A | B | C | D |
|---|---|---|---|---|
| 1 | **BMI Range** | | | **Category** |
| 2 | 0 | to | 18.4 | Underweight |
| 3 | 18.5 | to | 24.9 | Normal weight |
| 4 | 25 | to | 29.9 | Overweight |
| 5 | 30 | and bigger | | Obesity |

What is BMI? BMI means 'Body Mass Index'. It is a formula which indicates how fat someone is, in a fairly simple way. It is calculated by dividing the weight (kg) by the square of the height (m). By comparing the BMI with the BMI Ranges shown in the worksheet the BMI number can be used to show what weight category a person is in. For example, if a person's BMI is 17, then it is within the BMI range of 0 to 18.4 and the corresponding weight category is 'Underweight'. As you can see, cell A2 to cell D5 contains the BMI ranges together with the corresponding weight categories.

Using the worksheet shown above, three of the following statements are correct. One is not correct. Which one is not correct?

A) `=VLOOKUP(23, $A$2:$D$5, 4)` gives you 'Normal weight'
B) `=VLOOKUP(23, $A$2:$D$5, 4, FALSE)` gives you '#N/A'
C) `=VLOOKUP(23, $A$2:$D$5, 1)` gives you '23'
D) `=VLOOKUP(-5, $A$2:$D$5, 4)` gives you '#N/A'

**Answer (A/B/C/D):** _____

Q4)

| | A | B | C | D |
|---|---|---|---|---|
| 1 | **Victims of Telephone Scams in Hong Kong** | | | |
| 2 | **Reported** | **Gender** | **Job** | **Amount Lost** |
| 3 | 2018 | F | Runs a real estate investment company | HKD 180,000,000 |
| 4 | 2018 | F | A supervisor in the accounts department of a local trading company | HKD 26,400,000 |
| 5 | 2016 | M | Businessman | HKD 58,000,000 |
| 6 | 2016 | F | City U Professor | HKD 781,000 |
| 7 | 2016 | M | HKUST Professor | HKD 640,000 |
| 8 | 2015 | M & F | Famous Musician & Singer | HKD 22,800,000 |
| 9 | 2015 | M | Senior Executive with mainland firm | HKD 15,300,000 |
| 10 | 2015 | M | Senior manager of a global investment bank | HKD 6,300,000 |
| 11 | | | | |
| 12 | | | *Enter a year here to check if there were any scams in that year:* | 2017 |
| 13 | | | *The answer is:* | No |

In the space below, write a suitable formula for cell D13. The formula must show "Yes" if the year entered in D12 is shown in the first column. Otherwise, the formula must show "No".

**Answer:** _____

Q5)

The rules for *precedence* in VBA (and other programming languages) are the same as the rules for precedence in cell formulas which we looked at.

Which one of the following VBA code gives a different result compared to the other three?

A) `MsgBox (10 + 10) / 2`
B) `MsgBox (10 / 2 + 5)`
C) `MsgBox (10 + 10) / (4 / 2)`
D) `MsgBox (2 * 10) / 4 / 2`

**Answer (A/B/C/D): _____**

**Answers to Week 3 Review Questions**

Q1)
The answer is:
`=VLOOKUP(B1, A5:B11, 2, FALSE)`

The name mentioned in cell B1 should be searched in the first column of the range A5:B11. So the first blank should be A5:B11. The second blank is 2 because the corresponding value in the second column (within A5:B11) of the same row should be returned. Finally, it should be noted that the names in the cells A5:A11 are not sorted, which means approximate matching won't work properly. So exact matching should be employed. That means the third blank should be `FALSE`, because `FALSE` means Excel will look for an exact match.

Q2)
The answer is:
`Poor Dad!`

The condition in the IF formula is `VLOOKUP("Dad", A2:B5, 2, FALSE) < MAX(B2:B5)`. Since the fourth parameter in `VLOOKUP("Dad", A2:B5, 2, FALSE)` is `FALSE`, it looks for an exact match of the value in the first parameter (`"Dad"`) and returns the corresponding value in the second column of the same row (within A2:B5). Therefore, it returns the money that `"Dad"` has which is 20. `MAX(B2:B5)` finds the maximum value among cell B2 to cell B5 which is 500. Since 20 is less than (<) 500, the condition is `TRUE`, and the true part `"Poor Dad!"` is returned.

Q3)
The answer is C.

Choice A is correct. The formula `=VLOOKUP(23, $A$2:$D$5, 4)` performs approximate matching among the values in cell A2 to cell A5. Since the numbers in cell A2 to cell A5 are sorted in increasing order, the closest value smaller than 23 is cell A3. Returning the 4th column in row 3 gives the value '`Normal weight`'.
Choice B is correct. The formula `=VLOOKUP(23, $A$2:$D$5, 4, FALSE)` performs exact matching among the values in cell A2 to cell A5. It is because the fourth parameter is `FALSE` and `FALSE` means Excel will look for an exact match. The number 23 does not exist in the conversion table, and '`#N/A`' error is returned.
Choice C is incorrect, as the formula `=VLOOKUP(23, $A$2:$D$5, 1)` performs approximate matching among the values in cell A2 to cell A5. Since the numbers in cell A2 to cell A5 are sorted in increasing order, the closest value smaller than 23 is cell A3. Returning the 1st row means returning the number in cell A3. The result is `18.5` instead of `23`.
Choice D is correct. The formula `=VLOOKUP(-5, $A$2:$D$5, 4)` performs approximate matching among the values in cell A2 to cell A5 which are sorted in increasing order. There isn't a number smaller than -5, therefore, '`#N/A`' error is returned.

Q4)
One possible answer is this:

```
=IF( ISNA(VLOOKUP(D12, A:A, 1, FALSE)), "No", "Yes")
```

The idea of this formula is to first try to search for the year in column A using VLOOKUP. If the year is not there then the lookup will fail, so the answer of the whole formula will be "No". Otherwise, the answer of the whole formula will be "Yes".

There are some possible variations of this answer, and there are other possible answers.

Q5)
The answer is D.

In choice A, `(10 + 10) / 2` evaluates to 10. As a result, 10 is the text shown in the message box.
In choice B, `(10 / 2 + 5)` evaluates to 10 as `10 / 2` is evaluated first because of the higher precedence level of the division operator (/) than that of the addition operator (+). As a result, 10 is the text shown in the message box.
In choice C, `(10 + 10) / (4 / 2)` evaluates to 10 as the parts in the brackets, `(10 + 10)` and `(4 / 2)`, are evaluated first. As a result, 10 is the text shown in the message box.
In choice D, `(2 * 10) / 4 / 2` evaluates to 2.5. The part in the brackets `(2 * 10)` is evaluated first which gives the value 20. Then, the division operators (/) are evaluated from left to right. The value 20 is divided by 4 first and then divided by 2. As a result, 2.5 is the text shown in the message box.